

Enhanced Shewhart Plots Using a Graphics Template and Annotation

Kathy Claude, Strong Memorial Hospital, Rochester, New York
Joseph Guido, Strong Memorial Hospital, Rochester, New York

ABSTRACT:

The Quality Reporting System (QRS) was designed to measure and monitor quality of services and resource utilization within Strong Memorial Hospital. Fifteen indicators for nine departments were initially identified to be followed. Each indicator for each department required a statistical process control chart (SPC), an interpretation section and a comment section. To reduce the amount of paper required to produce the report it was decided to consolidate the three parts of the report onto a single piece of paper per department indicator. This was accomplished by using the GREPLAY procedure to combine a Shewhart plot, an interpretation section and a workspace section into a single graphics output. The ANNOTATE FACILITY was used to create the interpretation and comment sections and to enhance the display of out of control points on the Shewhart plot.

INTRODUCTION:

The intent of the QRS was to monitor key outcomes measures for quality of services and resource utilization. Indicators include length of stay, mortality and readmission rates, volumes, customer satisfaction and cost measures. The data is plotted monthly and the report is distributed quarterly. The report needed to deliver a lot of information but in a concise and simple to understand format. To keep the size of the report to a minimum we limited the report to one sheet of paper per indicator per department. Each sheet of paper was divided into three sections using a SAS/GRAPH® template with the TREPLAY statement in PROC GREPLAY. The largest section contains the SPC chart constructed with SAS/QC®. Two passes of the SHEWHART procedure were required. The first pass identifies the statistically out of control points, defined by the Western Electric Tests. A data set was created in order to label the alert points with arrows as defined by our customers. This was done with the ANNOTATE FACILITY. The second pass of PROC SHEWHART produces the annotated SPC chart. The second section contains four bulleted interpretation points created with the GSLIDE

procedure and ANNOTATE. Since the QRS was intended to be a working document the third section was designed as a workspace area for individual departments to comment on targets and actions for each indicator. This was also created using PROC GSLIDE and ANNOTATE.

IDENTIFY OUT OF CONTROL POINTS

PROC SHEWHART is run with NOCHART and OUTTABLE options to get the out of control points.

```
Proc shewhart history = sugi.sugidata
                    limits = sugi.xlimits;
xchart los*month / nochart
                    tests=1 to 8
                    test2run=8
                    outtable=alerts
                    stddeviations
                    alln
                    readlimits;

run;
```

THE ALERT ARROWS:

The alert arrows are created using the ANNOTATE FACILITY. The MARKER font is used to create the arrow. It is placed at the top of the graph over the month of occurrence. The size of the arrow is determined by when the alert occurred.

```
data anno;
length style text $8;
retain function 'label' xsys '2' ysys '5' hsys '4';
set alerts;
where _tests_ ne ' ';
x=month;
y=100;
style='marker';
position='B';
text='H';
if _n_ <22 then size=1;
else size=4;
run;
```

THE ANNOTATED SHEWHART PLOT

Re-run the shewhart plot adding the arrow annotation.

```

goptions hsize= vsize= hpos= vpos=;

proc shewhart history = sugi.sugidata
  gout = outgr1
  annotate=anno
  graphics
  limits=sugi.xlimits;
title1 font=swissb
  'Department of XXX Average Length of
  Stay';
title2 font=swiss
  'SMH Inpatient';
title3 font=swiss
  'January 1995 to December 1996';
title5 ' ';
label losx='Days';
format month monyy5.;
xchart los*month /          nochart2
                             test2=1 to 8
                             test2run=8
                             nolegend
                             nolimitslegend
                             nohlabel
                             skipphlabels=2
                             ndecimal=1
                             stddeviations
                             alln
                             testlabel=none
                             font=swiss
                             readlimits;

run;

```

CREATE THE INTERPRETATION AREA

A blank interpretation block is created using PROC GSLIDE and ANNOTATE. The block provides room for four bulleted interpretive comments with four lines of text each. The actual text is added at run time.

```

%annomac;

%macro
interp(txt1,txt2,txt3,txt4,txt5,txt6,txt7,txt8,txt9,txt10
,txt11,txt12,txt13,txt14,txt15,txt16);

goptions reset=global
  gunit=pct
  hsize=6 cm hpos=100
  vsize=12 cm vpos=40;

```

```

data anno2;
length style function $8 text $42;
retain xsys ysys hsys '5';
%rect(0,98,60,90, ,1,1);
%label(5,95,'Interpretation',1,0,0.00,4,swiss,6);
%rect(0,85,100,0, ,1,1);
%label(5,80,'J',1,0,0.00,3,special,6);
%label(10,80,"&txt1",1,0,0.00,2,swiss,6);
%label(10,76,"&txt2",1,0,0.00,2,swiss,6);
%label(10,72,"&txt3",1,0,0.00,2,swiss,6);
%label(10,68,"&txt4",1,0,0.00,2,swiss,6);
%label(5,59,'J',1,0,0.00,3,special,6);
%label(10,59,"&txt5",1,0,0.00,2,swiss,6);
%label(10,55,"&txt6",1,0,0.00,2,swiss,6);
%label(10,51,"&txt7",1,0,0.00,2,swiss,6);
%label(10,47,"&txt8",1,0,0.00,2,swiss,6);
%label(5,38,'J',1,0,0.00,3,special,6);
%label(10,38,"&txt9",1,0,0.00,2,swiss,6);
%label(10,34,"&txt10",1,0,0.00,2,swiss,6);
%label(10,30,"&txt11",1,0,0.00,2,swiss,6);
%label(10,26,"&txt12",1,0,0.00,2,swiss,6);
%label(5,17,'J',1,0,0.00,3,special,6);
%label(10,17,"&txt13",1,0,0.00,2,swiss,6);
%label(10,13,"&txt14",1,0,0.00,2,swiss,6);
%label(10,9,"&txt15",1,0,0.00,2,swiss,6);
%label(10,5,"&txt16",1,0,0.00,2,swiss,6);

```

```

proc gslide name='interp'
  annotate=anno2
  gout=outgr1;

run;
quit;
%mend interp;

%interp(txt1=The arrow at Dec 1996 shows
ALOS,
txt2=more than 3 std. deviations
below,
txt3=the mean.,
txt4=,
txt5=,
txt6=,
txt7=,
txt8=,
txt9=,
txt10=,
txt11=,
txt12=,
txt13=Note:,
txt14=Large arrow = New alert.,
txt15=Small arrow = Previous alert.,
txt16=Control limits are 3 SD from
the mean.);

```

CREATE THE WORKSPACE AREA

```
goptions hsize=6 cm hpos=100
        vsize=5 cm vpos=40;

data anno3;
length style function $8 text $45;
retain xsys ysys hsys '5';
%rect(0,100,40,90, ,1,1);
%label(5,95,'Comments',1,0,0.00,6,swiss,6);
%rect(0,85,100,0, ,1,1);
%label(5,80,'Council Set Targets:',1,0,0.00,4,
        swiss,6);
%line(5,69,95,69, ,1,3);
%line(5,61,95,61, ,1,3);
%line(5,53,95,53, ,1,3);
%label(5,42,'Actions Taken:',1,0,0.00,4,
        swiss,6);
%line(5,31,95,31, ,1,3);
%line(5,23,95,23, ,1,3);
%line(5,15,95,15, ,1,3);
run;

proc gslide name='workspace'
        annotate=anno3
        gout=outgr1;
run;
quit;
```

PUT IT ALL TOGETHER

Put the Shewhart plot, interpretive comments and the workspace together using PROC GREPLAY with the TREPLAY statement.

```
goptions hsize= hpos= vsize= vpos=;
```

```
proc greplay nofs
        igout=outgr1
        tc=tempcat
        gout=outgr1;
```

```
tdef qrscht
1/ llx=0      lly=0
   lrx=70     lry=0
   urx=70     ury=100
   ulx=0      uly=100
2/ llx=75     lly=35
   lrx=100    lry=35
   urx=100    ury=100
   ulx=75     uly=100
3/ llx=75     lly=0
   lrx=100    lry=0
   urx=100    ury=32
   ulx=75     uly=32;
```

```
template qrscht;
treplay 1:shewhart
```

```
2:interp
3:workspace;
```

```
run;
quit;
```

FUTURE ENHANCEMENTS:

Short term plans for the QRS include adding color to the graphs and printing on legal size paper to enhance the presentation for posting on department walls. We also plan to switch the programming platform from MVS mainframe to the SAS SYSTEM for PCs. This will allow us to use the editing features of the windowing environment to enter the interpretive comments, eliminating the need to hard code text in the program. Long term plans include an electronic version of the QRS with drill-down capabilities to examine data points in more detail.

CONCLUSION:

Customized SPC charts are easily produced using the ANNOTATE FACILITY and PROC GREPLAY with a TEMPLATE statement. Here we demonstrate adding alert arrows to out of control points and short interpretation and workspace sections to aid customers with their quality improvement efforts.

REFERENCES:

SAS Institute, Inc. (1995), *SAS/QC Software Usage and Reference, Version 6, First Edition*, Cary, NC: SAS Institute, Inc.

SAS Institute, Inc. (1990), *SAS/Graph Software: Reference, First Edition*, Cary, NC: SAS Institute, Inc.

SAS, SAS/GRAPH and SAS/QC are registered trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.


AUTHOR CONTACT:

Kathy Claude
Office of Clinical Practice Evaluation
Strong Memorial Hospital
601 Elmwood Avenue, Box 612
Rochester, New York 14642

Tel: (716) 275-0958
E-mail: kclaude@smhqm.urmc.rochester.edu

Department of XXX Average Length of Stay

SMH Inpatient
January 1995 to December 1996



Interpretation

- The arrow at Dec 1996 shows ALOS more than 3 std. deviations below the mean.
-
-
- Note:
 Large arrow = New alert.
 Small arrow = Previous alert.
 Control limits are 3 SD from the mean.

Comments

Council Set Targets:

Actions Taken:
